



핵심 예제로 배우는 아두이노 프로그래밍

Arduino Programming

연습문제
해답

1장

1장 문제 1 마이크로컨트롤러 v.s. 마이크로프로세서

컴퓨터의 중앙처리장치를 하나의 칩으로 만든 반도체 소자를 마이크로프로세서라 한다. 마이크로프로세서는 1969년 Intel에서 발표한 4비트 마이크로프로세서인 4004에서부터 시작되어 현재에 이르고 있으며, 현재 사용되고 있는 대부분의 중앙처리장치는 마이크로프로세서로 구현되고 있다.

마이크로프로세서 중 하나의 칩에 중앙처리장치의 기능뿐만이 아니라 일정 용량의 메모리와 입출력 인터페이스까지 하나의 칩으로 만든 반도체 소자를 마이크로컨트롤러라고 한다. 마이크로컨트롤러는 데스크톱 컴퓨터에서 메인 보드의 기능을 모두 포함하고 있는 것으로 볼 수 있으며, 단일 칩 마이크로컴퓨터 또는 마이컴이라고도 부른다.

마이크로프로세서가 범용의 목적을 가지는데 비해 마이크로컨트롤러는 제어용으로 사용할 목적으로 만들어져서 사용 목적 자체가 다르다. 마이크로프로세서가 컴퓨터에서 주로 사용된다면 마이크로컨트롤러는 낮은 성능과 싼 가격으로 임베디드 시스템에 주로 사용된다.

1장 문제 3 RISC v.s. CISC, 하버드 구조 v.s. 폰 노이만 구조

전통적으로 중앙처리장치는 CISC 구조를 바탕으로 한다. CISC는 상대적으로 많은 수의 기계어 명령어를 가지므로 명령어마다 처리 속도가 다르고, 회로가 복잡하며, 많은 전력을 소비하는 등의 단점이 있다. 반면, 명령어 처리를 위한 전용 하드웨어를 가질 수 있으므로 특정 명령어를 상대적으로 빨리 처리할 수 있다.

RISC는 CISC의 많은 명령어 중 자주 사용되는 일부 명령어만으로 구성된, 적은 명령어를 바탕으로 하는 시스템이다. 명령어의 수가 줄어듦에 따라 모든 명령어의 처리 속도는 동일하고, 회로가 간단하며, 상대적으로 적은 전력을 소비한다.

CISC 구조는 Intel로 대표되는 데스크톱 컴퓨터의 중앙처리장치에서 주로 사용되고 있는 반면, 모바일 기기와 마이크로컨트롤러에서 사용하는 대부분의 중앙처리장치는 RISC 구조를 바탕으로 하고 있다. 하지만 최근 CISC와 RISC는 서로의 장점을 받아들여 이전과 같이 경계가 명확하지는 않다.

폰 노이만 구조는 현대 컴퓨터의 근간이 되는 내장 메모리 순차처리 방식을 가리킨다. 폰 노이만 구조에서는 데이터와 프로그램이 구별 없이 메모리에 저장되고, 중앙처리장치로 순차적으로 옮겨진 후 실행된다. 하드웨어의 발전에 따라 중앙처리장치의 속도는 비약적으로 빨라졌지만 메모리의 속도는 그에 비해 느리다. 따라서 메모리에 위치하는 데이터와 프로그램을 중앙처리장치로 옮겨오는 시간이 중앙처리장치의 처리 시간보다 느려 병목현상이 발생하게 되었다. 이를 개선하기 위해 제안된 구조가 하버드 구조로, 하버드 구조에서는 데이터와 프로그램을 별도의 메모리에 저장하고 각각 별도의 경로(버스)를 통해 중앙처리장치로 옮겨진다. 폰 노이만 구조에 비해 하버드 구조는 회로가 복잡한 단점이 있지만 폰 노이만 구조에 비해 빠른 속도로 메모리 내의 정보를 중앙처리장치로 옮겨갈 수 있으므로 속도 측면에서는 유리하다.

2장

2장 문제 1

```
void setup()
{
    Serial.begin(9600);
}

void loop()
{
    byte data = 0b11110101;

    Serial.print("Bit Extraction : ");
    for(int i = 0; i < 8; i++){
        byte result = bitRead(data, i);
        Serial.print(result);
    }
    Serial.println();

    Serial.print("Bit Set      : ");
    for(int i = 4; i < 8; i++){
        bitSet(data, i);
    }
    Serial.println(data, BIN);

    Serial.print("Bit Inversion : ");
    for(int i = 0; i < 4; i++){
        bitWrite(data, i, bitRead(data, i) ^ 1);
    }
    Serial.println(data, BIN);

    Serial.print("Bit Clear     : ");
    for(int i = 0; i < 4; i++){
        bitClear(data, i);
    }
    Serial.println(data, BIN);

    while(true);
}
```

2장 문제 3

```
void setup()
{
    Serial.begin(9600);
}

void loop()
{
    int A = 20, B = 30, i;

    for(i = min(A, B); i > 0; i--){
        int remain1 = A % i;
        int remain2 = B % i;

        if(remain1 == 0 && remain2 == 0) break;
    }

    Serial.println(String("GCD is ") + i);

    while(true);
}
```

3장

3장 문제 1

```
String floatToString(float f, int n = 2){  
    String s = "";  
  
    s = String((int)f) + ".";  
    f = f - (int)f;  
  
    for(int i = 0; i < n; i++){  
        f = f * 10;  
        s = s + (int)f;  
        f = f - (int)f;  
    }  
    return s;  
}  
  
void setup(){  
    Serial.begin(9600);  
}  
  
void loop(){  
    float f = 3.14159;  
    String str = floatToString(f, 10);  
  
    Serial.println("=> " + str);  
  
    while(true);  
}
```

3장 문제 3

```
String buffer = "";

void setup(){
    Serial.begin(9600);
}

void loop(){
    while(Serial.available()){
        char ch = Serial.read();
        if(ch == '\r' || ch == '\n'){
            process_decimal();
            buffer = "";
            break;
        }
        else
            buffer += ch;
    }
}

void process_decimal(){
    int value = buffer.toInt();
    String converted = "";
    Serial.println("String buffer : " + buffer);

    while(value > 0){
        int remain = value % 5;
        value = value / 5;
        converted = remain + converted;
    }

    Serial.println(converted);
}
```

4장

4장 문제 1

```
int pins[] = {2, 3, 4, 5, 6, 7, 8, 11};

void setup() {
    Serial.begin(9600);
    for(int i = 0; i < 8; i++){
        pinMode(pins[i], INPUT);
    }
}

void loop() {
    int state = 0;
    for(int i = 0; i < 8; i++){
        boolean value = digitalRead(pins[i]);
        state = (state << 1) + value;

        if(value){
            Serial.print("1 ");
        } else{
            Serial.print("0 ");
        }
    }
    Serial.println(String(" : ") + state);

    delay(1000);
}
```

4장 문제 3

(생략)

5장

5장 문제 1

```
#include <pitches.h>

int speakerPin = 4;
int melody[] = {
    NOTE_E4, NOTE_G4, NOTE_G4, NOTE_E4, NOTE_G4, NOTE_G4,
    NOTE_A4, NOTE_A4, NOTE_A4, NOTE_A4, NOTE_A4, 0,
    NOTE_G4, NOTE_G4, NOTE_G4, NOTE_G4, NOTE_F4, NOTE_F4, NOTE_F4,
    NOTE_E4, NOTE_E4, NOTE_E4, NOTE_E4, NOTE_E4, 0,
    NOTE_E4, NOTE_G4, NOTE_G4, NOTE_G4, NOTE_E4, NOTE_G4, NOTE_G4,
    NOTE_A4, NOTE_A4, NOTE_E4, NOTE_E4, NOTE_G4, 0,
    NOTE_F4, NOTE_F4, NOTE_F4, NOTE_E4, NOTE_E4, NOTE_E4, NOTE_E4,
    NOTE_D4, NOTE_D4, NOTE_G4, NOTE_G4, NOTE_C4, 0
};

int one = 500, half = 250;

int duration[] = {
    half, half, one, half, half, one, half, half, half, half+one, half,
    half, half, half, half, half, half, half, half, half, half, half+one, half,
    half, half, half, half, half, half, one, half+(half/2), half/2, half, half, half+one, half,
    half, half, half, half, half, half, half, half, half, half, half, half+one, half
};

void setup(){
}

void loop(){
    for(int thisNote = 0; thisNote < sizeof(melody) / sizeof(int); thisNote++){
        tone(speakerPin, melody[thisNote], duration[thisNote]);

        delay(duration[thisNote]);
        noTone(speakerPin);
    }

    while(true);
}
```

5장 문제 3

```
int button = 9;
int led = 10;
boolean state = false;

void setup() {
    pinMode(button, INPUT);
    pinMode(led, OUTPUT);
}

void loop() {
    boolean value = digitalRead(button);

    if(value){
        delay(200);
        state = !state;
    }

    digitalWrite(led, state);
}
```

6장

6장 문제 1

```
#include <pitches.h>
#define DURATION 10
int notes[] = { NOTE_C4, NOTE_D4, NOTE_E4, NOTE_F4,
    NOTE_G4, NOTE_A4, NOTE_B4, NOTE_C5};
int speakerPin = 4;
int analogPin = A3;

void setup(){
}

void loop()
{
    int val = analogRead(analogPin);
    int index = map(val, 0, 1023, 0, 7);
    tone(speakerPin, notes[index], DURATION);
}
```

6장 문제 3

```
int pins[] = {5, 6, 7, 8, 9, 10, 11, 12};
int analog_pin = A3;

void setup(){
    for(int i = 0; i < 8; i++){
        pinMode(pins[i], OUTPUT);
    }
}

void loop(){
    int value = analogRead(analog_pin);
    int LED_position = map(value, 0, 1023, 0, 7);

    for(int i = 0; i < 8; i++){
        if(i == LED_position)
            digitalWrite(pins[i], HIGH);
        else
            digitalWrite(pins[i], LOW);
    }
}
```

7장

7장 문제 1

```
int pins[] = {5, 6, 7, 8, 10, 11, 12, 13};  
String buffer;  
boolean process;  
  
void setup(){  
    for(int i = 0; i < 8; i++){  
        pinMode(pins[i], OUTPUT);  
    }  
  
    Serial.begin(9600);  
    buffer = "";  
    process = false;  
}  
  
void loop(){  
    while(Serial.available() > 0){  
        char data = Serial.read();  
        if(data == '\n'){  
            process = true;  
            break;  
        }  
        else{  
            buffer = String(data) + buffer;  
        }  
    }  
  
    if(process){  
        int number = 0;  
        boolean binarySequence = true;  
        for(int i = 0; i < buffer.length(); i++){  
            if(buffer[i] != '0' && buffer[i] != '1'){  
                binarySequence = false;  
                break;  
            }  
            number = number + (buffer[i] - '0') * (1 << i);  
        }  
  
        process = false;  
        buffer = "";  
  
        if(!binarySequence)  
            Serial.println("Not a valid binary number sequence !!");  
        else{  
            Serial.println("I got " + String(number));  
            if(number < 0 || number > 255){  
                Serial.println("Wrong number !!!");  
            }  
            else{  
                for(int i = 0; i < 8; i++){  
                    boolean on_off = bitRead(number, i);  
                    digitalWrite(pins[i], on_off);  
                }  
            }  
        }  
    }  
}
```

7장 문제 3

```
#define INTERVAL1 500
#define INTERVAL2 1000

int LED1 = 10, LED2 = 13;
unsigned long start1, start2, current1, current2;
boolean state1, state2;

void setup()
{
    pinMode(LED1, OUTPUT);
    pinMode(LED2, OUTPUT);

    start1 = millis();
    start2 = millis();

    state1 = false;
    state2 = false;

    digitalWrite(LED1, state1);
    digitalWrite(LED2, state2);
}

void loop()
{
    current1 = millis();
    if(current1 - start1 > INTERVAL1){
        start1 = current1;
        state1 = !state1;
        digitalWrite(LED1, state1);
    }

    current2 = millis();
    if(current2 - start2 > INTERVAL2){
        start2 = current2;
        state2 = !state2;
        digitalWrite(LED2, state2);
    }
}
```

8장

8장 문제 1

```
byte patterns[] = {  
    0xFC, 0x60, 0xDA, 0xF2, 0x66, 0xB6, 0xBE, 0xE4, 0xFE, 0xE6  
};  
  
int pins[] = {5, 6, 7, 8, 10, 11, 12, 13};  
  
void setup(){  
    Serial.begin(9600);  
  
    for(int i = 0; i < 8; i++){  
        pinMode(pins[i], OUTPUT);  
        digitalWrite(pins[i], LOW);  
    }  
}  
  
void loop(){  
    char input;  
  
    if(Serial.available()){  
        input = Serial.read();  
        Serial.println(String("You typed ") + input + "");  
  
        if(input >= '0' && input <= '9'){  
            for(int i = 0; i < 8; i++){  
                boolean on_off = bitRead(patterns[input - '0'], i);  
                digitalWrite(pins[7 - i], on_off);  
            }  
        }  
        else{  
            for(int i = 0; i < 8; i++){  
                digitalWrite(pins[i], HIGH);  
            }  
        }  
    }  
}
```

8장 문제 3

```
byte patterns[] = {  
    0xFC, 0x60, 0xDA, 0xF2, 0x66, 0xB6, 0xBE, 0xE4, 0xFE, 0xE6  
};  
  
int pins[] = {5, 6, 7, 8, 10, 11, 12, 13};  
int index = 0;  
int button_pin = 9;  
  
void setup(){  
    for(int i = 0; i < 8; i++){  
        pinMode(pins[i], OUTPUT);  
    }  
    pinMode(button_pin, INPUT);  
  
    for(int i = 0; i < 8; i++){  
        boolean on_off = bitRead(patterns[index], i);  
        digitalWrite(pins[7 - i], on_off);  
    }  
}  
  
void loop(){  
    boolean pressed = digitalRead(button_pin);  
    if(pressed){  
        delay(200);  
        index = (index + 1) % 10;  
    }  
  
    for(int i = 0; i < 8; i++){  
        boolean on_off = bitRead(patterns[index], i);  
        digitalWrite(pins[7 - i], on_off);  
    }  
}
```

9장

9장 문제 1

```
byte patterns[] = {  
    0xFC, 0x60, 0xDA, 0xF2, 0x66, 0xB6, 0xBE, 0xE4, 0xFE, 0xE6  
};  
  
int digit_select_pin[] = {12, 13, A0, A1};  
int segment_pin[] = {2, 3, 5, 6, 7, 8, 10, 11};  
  
int SEGMENT_DELAY = 5;  
int count = 20;  
boolean toggle = false;  
  
unsigned long time_previous, time_current;  
  
void setup(){  
    for(int i = 0; i < 4; i++){  
        pinMode(digit_select_pin[i], OUTPUT);  
    }  
    for(int i = 0; i < 8; i++){  
        pinMode(segment_pin[i], OUTPUT);  
    }  
  
    time_previous = millis();  
}  
  
void show_digit(int pos, int number){  
    for(int i = 0; i < 4; i++){  
        if(i + 1 == pos)  
            digitalWrite(digit_select_pin[i], LOW);  
        else  
            digitalWrite(digit_select_pin[i], HIGH);  
    }  
  
    for(int i = 0; i < 8; i++){  
        boolean on_off = bitRead(patterns[number], 7 - i);  
        digitalWrite(segment_pin[i], on_off);  
    }  
}  
  
void clear_digit(int pos){  
    for(int i = 0; i < 8; i++){  
        digitalWrite(segment_pin[i], LOW);  
    }  
}  
  
void loop(){  
    time_current = millis();  
  
    if(time_current - time_previous >= 500){  
        time_previous = time_current;  
        toggle = !toggle;  
        if(!toggle) count--;  
        if(count == -1) count = 20;  
    }  
  
    if(count < 10){
```

```
if(toggle){
    show_digit(4, count);
    delay(SEGMENT_DELAY);
}
else{
    clear_digit(4);
    delay(SEGMENT_DELAY);
}
else{
    show_digit(3, count / 10);
    delay(SEGMENT_DELAY);
    show_digit(4, count % 10);
    delay(SEGMENT_DELAY);
}
}
```

9장 문제 3

```
byte patterns[] = {
    0xFC, 0x60, 0xDA, 0xF2, 0x66, 0xB6, 0xBE, 0xE4, 0xFE, 0xE6
};

int digit_select_pin[] = {12, 13, A0, A1};
int segment_pin[] = {2, 3, 5, 6, 7, 8, 10, 11};
int SEGMENT_DELAY = 5;
int count = 0;
int pos = 1;

unsigned long time_previous, time_current;

void setup(){
    for(int i = 0; i < 4; i++){
        pinMode(digit_select_pin[i], OUTPUT);
    }
    for(int i = 0; i < 8; i++){
        pinMode(segment_pin[i], OUTPUT);
    }

    time_previous = millis();
}

void show_digit(int pos, int number){
    for(int i = 0; i < 4; i++){
        if(i + 1 == pos)
            digitalWrite(digit_select_pin[i], LOW);
        else
            digitalWrite(digit_select_pin[i], HIGH);
    }

    for(int i = 0; i < 8; i++){
        boolean on_off = bitRead(patterns[number], 7 - i);
        digitalWrite(segment_pin[i], on_off);
    }
}

void loop(){
    time_current = millis();

    if(time_current - time_previous >= 250){
        time_previous = time_current;

        pos++;
        if(pos == 5){
            pos = 1;
            count = (count + 1) % 10;
        }
    }

    show_digit(pos, count);
}
```

10장

10장 문제 1

```
byte patterns[] = {  
    0b10000000,  
    0b01000000,  
    0b00100000,  
    0b00010000,  
    0b00001000,  
    0b00000100 };  
  
int dataPin = 11;  
int latchPin = 12;  
int clockPin = 13;  
  
int index = -1;  
  
void setup() {  
    pinMode(dataPin, OUTPUT);  
    pinMode(clockPin, OUTPUT);  
    pinMode(latchPin, OUTPUT);  
}  
  
void loop() {  
    index = (index + 1) % 6;  
  
    digitalWrite(latchPin, LOW);  
    shiftOut(dataPin, clockPin, LSBFIRST, ~patterns[index]);  
    digitalWrite(latchPin, HIGH);  
  
    delay(500);  
}
```

10장 문제 3

```
byte pattern = 0B01100000;

int dataPin = 11;
int latchPin = 12;
int clockPin = 13;

boolean move_right = false;

void setup() {
    pinMode(dataPin, OUTPUT);
    pinMode(clockPin, OUTPUT);
    pinMode(latchPin, OUTPUT);

    Serial.begin(9600);
}

byte circular_right_shift(byte d, int n)
{
    byte d1 = d >> n;
    byte d2 = d << (8 - n);

    return (d1 | d2);
}

byte circular_left_shift(byte d, int n)
{
    byte d1 = d >> (8 - n);
    byte d2 = d << n;

    return (d1 | d2);
}

void loop() {
    if(move_right){
        pattern = circular_left_shift(pattern, 1);
    }
    else{
        pattern = circular_right_shift(pattern, 2);
    }
    Serial.println(pattern, BIN);

    move_right = !move_right;

    digitalWrite(latchPin, LOW);
    shiftOut(dataPin, clockPin, MSBFIRST, pattern);
    digitalWrite(latchPin, HIGH);

    delay(500);
}
```

11장

11장 문제 1

```
#include <Wire.h>
#include <Adafruit_MCP23017.h>

Adafruit_MCP23017 mcp;

int pattern_no = 0;

void setup() {
    mcp.begin();

    for(int i = 0; i < 16; i++){
        mcp.pinMode(i, OUTPUT);
    }
}

void loop() {
    int data = 1 << pattern_no;

    mcp.writeGPIOB(data);

    pattern_no = (pattern_no + 1) % 16;

    delay(100);
}
```

11장 문제 3

```
#include <Wire.h>
#include <Adafruit_MCP23017.h>

Adafruit_MCP23017 mcp;

void setup() {
    mcp.begin();

    for(int i = 0; i < 16; i++)
        mcp.pinMode(i, OUTPUT);
}

void loop() {
    mcp.writeGPIOA(0xAA);
    mcp.writeGPIOB(0xAA);
    delay(500);
    mcp.writeGPIOA(0x55);
    mcp.writeGPIOB(0x55);
    delay(500);
}
```

12장

12장 문제 1

```
#include <LiquidCrystal.h>

LiquidCrystal lcd(2, 3, 5, 6, 7, 8);

void setup(){
    lcd.begin(16, 2);
    lcd.clear();
    randomSeed(analogRead(A5));
    Serial.begin(9600);
}

void loop(){
    int row = random(0, 2);
    int col = random(0, 17);

    char ch = random(0, 53);
    if(ch == 0) ch = ' ';
    else if(ch <= 26) ch = ch + 'A' - 1;
    else ch = ch + 'a' - 27;

    lcd.setCursor(col, row);
    lcd.write(ch);

    Serial.println(String("(") + row + ", " + col + ") " + ch);

    delay(100);
}
```

12장 문제 3

```
#include <LiquidCrystal.h>

LiquidCrystal lcd(2, 3, 5, 6, 7, 8);

byte user[][][8] = {
    {B00000, B00000, B00000, B00000, B00000, B00000, B10000, B00000},
    {B00000, B00000, B00000, B00000, B00000, B01000, B11000, B00000},
    {B00000, B00000, B00000, B00000, B00100, B01100, B11100, B00000},
    {B00000, B00000, B00000, B00010, B00110, B01110, B11110, B00000},
    {B00000, B00000, B00001, B00011, B00111, B01111, B11111, B00000}
};

void setup() {
    for(int i = 0; i < 5; i++){
        lcd.createChar(i, user[i]);
    }

    lcd.begin(16, 2);
    lcd.clear();
}

void loop() {
    int value = analogRead(A3);
    String value_string = String(value);

    int index = map(value, 0, 1023, 0, 4);
    lcd.home();
    lcd.write(byte(index));
    lcd.print(value);

    for(int i = 0; i < 4 - value_string.length(); i++){
        lcd.write(' ');
    }

    delay(100);
}
```

13장

13장 문제 1

```
int led_pin = 10;

void setup(){
    Serial.begin(9600);
    pinMode(led_pin, OUTPUT);
}

void loop(){
    int val = analogRead(A2);

    int pwm_value = map(val, 500, 900, 255, 0);
    pwm_value = constrain(pwm_value, 0, 255);

    Serial.println("Lumination : " + val + ", LED PWM : " + pwm_value);

    analogWrite(led_pin, pwm_value);

    delay(1000);
}
```

13장 문제 3

```
#define N 10

int values[N] = {0, };
int index = 0;
int count = 0;

void setup(){
    Serial.begin(9600);
}

void loop(){
    int val = analogRead(A2);
    values[index] = val;
    count++;

    index++;
    if(index == N) index = 0;

    float sum = 0.0;
    for(int i = 0; i < N; i++){
        sum += values[i];
    }

    if(count >= N) sum = sum / N;
    else sum = sum / count;
    Serial.println(sum);

    delay(1000);
}
```

14장

14장 문제 1

```
#include <EEPROM.h>

void setup(){
    Serial.begin(9600);
}

void write_float_EEPROM(int address, float value)
{
    byte *pointer = (byte *)&value;

    for(int i = 0; i < 4; i++)
        EEPROM.write(address + i, *(pointer + i));
}

float read_float_EEPROM(int address){
    float value = 0;
    byte *pointer = (byte *)&value;

    for(int i = 0; i < 4; i++)
        *(pointer + i) = EEPROM.read(address + i);

    return value;
}

void loop(){
    Serial.println("Write 12.34 to EEPROM");
    write_float_EEPROM(10, 12.34);

    Serial.println("Read from EEPROM");
    Serial.print(" >> ");
    Serial.println(read_float_EEPROM(10));

    while(true);
}
```

14장 문제 3

```
#include <Wire.h>
#include <Adafruit_MCP23017.h>
#include <Metro.h>

Adafruit_MCP23017 mcp;

Metro timer1(500);
Metro timer2(1000);

boolean toggle1 = true, toggle2 = true;

void setup(){
    mcp.begin();

    for(int i = 8; i < 16; i++){
        mcp.pinMode(i, OUTPUT);
    }
}

void loop(){
    if(timer1.check()){
        toggle1 = !toggle1;
        for(int i = 8; i < 12; i++)
            mcp.digitalWrite(i, toggle1);
    }
    if(timer2.check()){
        toggle2 = !toggle2;
        for(int i = 12; i < 16; i++)
            mcp.digitalWrite(i, toggle2);
    }
}
```

15장

15장 문제 1

```
#include <SoftwareSerial.h>

SoftwareSerial BTSerial(2, 3);

void setup()
{
    BTSerial.begin(9600);
    Serial.begin(9600);
}

void loop()
{
    int value = analogRead(A3);

    BTSerial.println(value);
    Serial.println(value);

    delay(1000);
}
```

15장 문제 3

```
#include <SoftwareSerial.h>

SoftwareSerial BTSerial(2, 3);
int button = 9;
int temp_pin = A1;

void setup()
{
    BTSerial.begin(9600);
}

void loop()
{
    if(digitalRead(button)){
        int input_value = analogRead(temp_pin);
        float input_voltage = 5.0 * input_value / 1023.0;
        float temperature = input_voltage * 100.0;

        BTSerial.println("Event occurred...");
        BTSerial.print(" ** Current temperature is ");
        BTSerial.println(temperature);

        delay(300);
    }
}
```